
Introduction
to lazy evaluation

Introduction to lazy evaluation
l A lazy program is a functional program that executes 

in “by-need” fashion
l Nothing is computed until it is “needed” 

l Here is a simple example:
fun lazy {LazyAdd X Y}

X+Y
end
S={LazyAdd 10 20}
{Browse S}

l Nothing is executed until S is needed:
% Displaying the addition S+100 needs S:
{Browse S+100}

Semantics
of lazy evaluation

Semantics of LazyAdd
l How does LazyAdd work?

l Semantics of a program is defined by translation into kernel language
l We will define what “needing a value” means

l We translate into kernel language:
proc {LazyAdd X Y R}

thread
{WaitNeeded R} R=X+Y

end
end

l The {WaitNeeded R} waits until another thread needs R to continue
l More precisely, it waits until another thread does {Wait R}
l This is part of dataflow execution…

l6

Dataflow semantics
l To understand WaitNeeded, 

we first recall how dataflow 
execution works

l Given any expression:
S=X+Y

l This is translated as:
local V in

{Wait X}
{Wait Y}
{PrimitiveAdd X Y V}
{Bind S V}

end

l This gives a dataflow execution:
l {Wait X} suspends until X is bound
l {Bind X V} binds X to V

l Programmer-accessible operations 
are defined using Wait, Bind, and a 
primitive operation:
l Arithmetic, boolean expressions
l Case statements
l Any operation with an input

l Function call {F X} where F must be 
bound to a function value

l Dot operation R.name where R  must 
be bound to a record

l {WaitNeeded X} suspends until 
another thread does {Wait X}

l7

Another example
l We use WaitNeeded directly:

declare X in
{WaitNeeded X}
X=100

l This displays an unbound variable:
{Browse X}

l This displays 100 twice (!):
{Browse X+0}

l8

General translation scheme
l Given any lazy function:

fun lazy {F X1 … Xn}
<expr>

end

l This is translated into:
proc {F X1 … Xn R}

thread
{WaitNeeded R} R=<expr>

end
end

l This translation gives the semantics, not the implementation!
l A compiler is free to optimize it while respecting the semantics

l9

Three kinds of 
producer-consumer

Producer-consumer code
l We give the code of a simple 

producer-consumer
l We will show three different 

ways to run the same code
l All three ways are declarative 

and end up with the same result

l Technically we are just taking 
advantage of the Church-
Rosser theorem
l All reduction orders of a lambda 

expression give the same result
l Also called confluence

fun {Prod L H}
{Delay 1000} % Wait 1000 ms
if L>H then nil
else L|{Prod L+1 H}
end

end

fun {Cons S Acc}
case S of H|T then

Acc+H|{Cons T Acc+H}
[] nil then nil
end

end

l11

Sequential execution
l We generate 10 elements

l Nothing is displayed until after 10 seconds
l This is a batch execution

declare S1 S2 in
{Browse S1}
{Browse S2}
S1={Prod 1 10}
S2={Cons S1 0}

l12

Concurrent execution
l We execute both calls in their own threads

l This is running deterministic dataflow (eager)
l What is the difference with the sequential version?

l This is an incremental execution

declare S1 S2 in
{Browse S1}
{Browse S2}
thread S1={Prod 1 10} end
thread S2={Cons S1 0} end

l13

Lazy execution
fun lazy {Prod L H}

{Delay 1000}
if L>H then nil
else L|{Prod L+1 H}
end

end

fun lazy {Cons S Acc}
case S of H|T then

Acc+H|{Cons T Acc+H}
[] nil then nil
end

end

l We annotate both functions as “lazy”
l We execute it:

declare S1 S2 in
{Browse S1}
{Browse S2}
S1={Prod 1 10}
S2={Cons S1 0}

l What is going on?
l Why is nothing computed?
l How do we run this?
l {Browse S2.2.1} displays the second

element of S2, which will activate its 
computation

l14

Eager versus lazy streams
l One way to understand the difference between eager 

and lazy is to see which agent is driving the execution

l In an eager stream, it is the producer that determines 
when elements are sent
l Termination is decided by the producer

l In a lazy stream, it is the consumer that determines 
when elements are sent
l Termination is decided by the consumer

l15

Infinite lists

Infinite lists
l With lazy evaluation we can compute with infinite loops

l We can write programs with infinite lists
l It works because the execution will only generate the elements 

that are needed
l An infinite list of integers starting with N:

fun lazy {Ints N} N|{Ints N+1} end
l Calling {Ints 1} displays an unbound variable:

L={Ints 1} {Browse L}
l We can force a computation by examining the list L:

{Browse L.1}
{Browse L.2.1}

l17

Semantics of infinite lists
l We can see how infinite lists work by translating to 

kernel language:
proc {Ints N R}

thread
{WaitNeeded R} R=N|{Ints N+1}

end
end

l When we need R by doing {Browse R.1}, this causes R 
to be bound to N|{Ints N+1}
l This causes one element of R to be computed
l The recursive call will immediately suspend again

l18

Forcing a computation
l We can force the evaluation of N elements of a list by 

traversing the list:
proc {Touch L N}

if N==0 then skip
else {Touch L.2 N-1} end

end

l This strange procedure does nothing by itself, yet it 
forces the work to be done:
{Touch L 10}
{Touch L 20}

l19

Hamming problem

l20

Hamming problem
l Richard Hamming (1915-1998) was an engineer and mathematician 

who worked at Bell Labs and invented many useful things
l Hamming codes, Hamming window, Hamming distance, etc.
l The Art of Doing Science and Engineering: Learning to Learn, by 

Richard Hamming, 1997.  This book is highly recommended!

l Today we will investigate the Hamming problem, a simple problem 
in number sequences
l It  is a dynamic problem where we do not know in advance how much 

needs to be computed → perfect for lazy evaluation!
l We will use lazy evaluation to design a simple and efficient solution to 

this problem

l21

Hamming problem
l Problem statement:

l Given the set of numbers of the form 2a 3b 5c with integers a,b,c ≥ 0
l It is asked to compute these numbers in increasing order: 1 | 2 | 3 | …

l We do not know in advance how many numbers of this sequence 
will be needed
l The program should let us compute them incrementally until we are 

satisfied
l The program should be efficient in time and memory!

l22

Algorithm idea

l We can program this with lazy lists

H = 1 | 2 | 3 | X |… § Idea: The next number X is 2 times,
3 times, or 5 times one of the 
previous numbers in the sequence

§ We need to keep three sequences 
derived from H, namely 2H, 3H and 
5H, and take the least number not 
yet used

§ Numbers 2 and 3 are already taken
§ Next number is either 4, 6, or 5
§ We take the minimum of these 

three: the next number is 4

2H = 2 | 4 | 6 | …

3H = 3 | 6 | 9 | …

5H = 5 | 10 | 15 | …

⟹ X = min(4,6,5) = 4

l23

Hamming program operations
l The algorithm needs two operations

l Multiply list elements by an integer
l Merge two ordered lists

l L2={Times L1 N}
l Each element of L2 is N times the element of L1

l L={Merge L1 L2}
l Assume L1 and L2 are in increasing order
l L contains elements of L1 and L2 in increasing order

l24

Hamming program
fun lazy {Times S N}

case S of H|T then
N*H|{Times T N}

end
end

fun lazy {Merge S1 S2}
case S1|S2 of (H1|T1)|(H2|T2) then

if H1<H2 then H1|{Merge T1 S2}
elseif H1>H2 then H2|{Merge S1 T2}
else /* H1==H2 */ H1|{Merge T1 T2}
end

end
end

l Main expression:
H=1|{Merge 

{Times H 2} 
{Merge {Times H 3} 

{Times H 5}}}
{Browse H}

l25

Lazy suspensions

l26

Lazy suspensions
l We defined lazy evaluation using threads and 

WaitNeeded
l This is correct but it does not show the execution

l Let us show the execution of a lazy program with a 
graphical approach

l We introduce the concept of lazy suspension:
Executing: L2={Times L1 3}
Creates a suspension: L2 → {Times L1 3}
“A thread is suspended on L2 that contains the body of {Times L1 3}”

l27

Execution of Hamming program

l Running the program creates five lazy suspensions
l The lazy suspension {Merge T2 M2} waits on M1
l Executing M1.1 activates the lazy suspension {Merge T2 M2}, 

which executes the body of {Merge T2 M2}, which then 
activates {Merge T3 T5} and {Times H 2}, and so forth!
l All five lazy suspensions are activated and five new ones are created
l At the end, M1.1 is bound to 2|M1’ with the new variable M1’

H=1|M1
{Merge T2 M2}

{Times H 2}

{Merge T3 T5}

{Times H 3}

{Times H 5}

l28

First activation: {Merge T2 M2}
l Request the second element of H:

{Browse M1.1}

l This activates {Merge T2 M2}:
l The body is executed:

case T2|M2 of (H1|T2’) | (H2|M2’) then
…

end
l The case needs the first elements of T2 and M2

l This activates {Times H 2} and {Merge T3 T5}
l The case waits patiently until T2 and M2 are bound

Activate
{Times H 2}

Activate
{Merge T3 T5}

l29

Next activations
l {Times H 2} and {Merge T3 T5} are activated

l The body of {Times H 2} is executed
l This binds T2=2|T2’ and creates a new lazy suspension on T2’: 

{Times M1 2}
l The body of {Merge T3 T5} is executed

l This activates {Times H 3} and {Times H  5}
l After executing these two functions, this binds M2=3|M2’ and creates 

a new lazy suspension on M2’: {Merge T3’ T5}

l Now the case in {Merge T2 M2}, which was waiting 
patiently, can be executed:

l It returns 2|M1’ with M1’={Merge T2’ M2} and creates a new lazy 
suspension on M1’

l30

Overall execution flow

M1.1 

{Merge T2 M2}

case T2|M2 ⋯ then
⋯

end

{Times H 2}

{Merge T3 T5}

case T3|T5 ⋯ then
⋯

end

{Times H 3}

{Times H 5}

activate

execute

bind M1

activate

bind T2
activate

execute
bind M

2
ac

tiv
ate

bin
d T

3

activate
bind T5

{Browse M1.1}

§ Doing M1.1 starts it all
§ Five lazy suspensions

are activated in a chain
§ Each one executes the

body of a function, which
returns result and creates
a new lazy suspension

§ M1 is bound at the end
so M1.1 can complete

l31

Lazy deterministic 
dataflow

l32

Five functional paradigms
l So far we have shown four paradigms of functional programming:

l Sequential functional programming
l Sequential functional programming with single assignment

l Allows data structures with “holes”, e.g., list functions are tail-recursive
l Deterministic dataflow

l Adds threads and dataflow synchronization
l Allows concurrent programming with streams

l Lazy evaluation
l Adds by-need evaluation (with WaitNeeded), where functions are executed 

only when their results are needed
l Allows programming with infinite lists

l There is a fifth paradigm:
l Lazy deterministic dataflow

l Adds both threads and lazy functions

l33

Lazy deterministic dataflow
l Lazy deterministic dataflow is the most powerful declarative 

paradigm:
l It has the strong properties of functional programming: confluence

and higher-order
l It has concurrency: independent activities which can get out of step 

with each other
l It has lazy evaluation: by-need computations which are only done 

when their result is needed

l What can we do with all this power?
l We give one example of a program that can be written in lazy 

deterministic dataflow, but not in any weaker paradigm

l34

Bounded buffer

l35

Bounded buffer (1)

l A producer-consumer pipeline has performance problems
l Variations in producer and consumer speeds can cause the 

system to perform poorly
l When a producer creates elements too quickly, the consumer 

cannot use the elements so the producer idles
l When a consumer needs more elements, the producer may not be 

able to produce them so the consumer idles

Producer Consumer
Stream

l36

Bounded buffer (2)

l Inserting a bounded buffer can solve these problems
l When the producer creates elements too quickly to be 

consumed, they are stored in the bounded buffer
l When the consumer needs more elements than can be 

produced, they are taken from the bounded buffer
l This improves performance by smoothing out fluctuations in 

producer and consumer speeds

Producer ConsumerBounded
buffer

l37

Bounded buffer (3)

l A bounded buffer fits in between a lazy producer and a lazy consumer
l The code of the producer and consumer is unchanged

l To the producer, the bounded buffer looks like a consumer
l To the consumer, the bounded buffer looks like a producer

l The bounded buffer “consumes” elements even when the consumer 
does not ask for them, and “produces” elements even when the 
producer does not make them

Producer
(lazy)

Consumer
(lazy)

Bounded
buffer

l38

Defining the bounded buffer
l Assume we have a producer-consumer pipeline:

thread S={Producer …} end
thread {Consumer S} end

l The bounded buffer is inserted in between:
thread S1={Producer …} end
thread {BoundedBuffer S1 S2 10} end
thread {Consumer S2} end

l We define the bounded buffer step-by-step
l We define the procedure {BoundedBuffer S1 S2 N} where S1 is the input 

stream, S2 is the output stream, and N is the buffer size
l We build the procedure in four steps, to make it easier to understand

l39

First step: pass elements
l The buffer outputs the same elements as it inputs:

proc {BoundedBuffer S1 S2 N}
fun lazy {Loop S1}

case S1 of H1|T1 then H1|{Loop T1} end
end

in
S2={Loop S1}

end

l40

Second step: startup
l The buffer asks for N elements on startup:

proc {BoundedBuffer S1 S2 N}
fun lazy {Loop S1}

case S1 of H1|T1 then H1|{Loop T1} end
end
End

in
End={List.drop S1 N} % Asking must not be lazy!
S2={Loop S1}

end

l {List.drop L N} is a library function that removes the first 
N elements from a list L

l41

Third step: staying full
l Whenever the consumer gets an element, the buffer asks for 

another element from the producer:

proc {BoundedBuffer S1 S2 N}
fun lazy {Loop S1 End}

case S1 of H1|T1 then H1|{Loop T1 End.2} end
end
End

in
End={List.drop S1 N}
S2={Loop S1 End}

end

l42

Fourth step: no blocking
l To avoid blocking the buffer’s main loop, both asks 

must be done in their own threads:

proc {BoundedBuffer S1 S2 N}
fun lazy {Loop S1 End}

case S1 of H1|T1 then
H1|{Loop T1 thread End.2 end} 

end
end
End

in
thread End={List.drop S1 N} end
S2={Loop S1 End}

end

In declarative programming,
threads are your friends!  
They are efficient. They can 
be added at will without 
adding bugs. They remove 
blocking and make the 
program more incremental.

All list functions, including
List.drop, work correctly
when used concurrently

l43

Example execution
l We create a pipeline with 

producer, bounded buffer, 
and consumer:

declare S1 S2 S3 in
{Browse S1}
{Browse S2}
{Browse S3}
S1={Prod 1 10}
{BoundedBuffer S1 S2 3}
S3={Cons S2 0}

l Note that the producer 
immediately produces 3 
elements, which are stored in 
the buffer

l When we consume one 
element, the buffer asks the 
producer for one element
l The buffer tries to stay full

l The buffer is eager until it is full, 
and then it becomes lazy

l44

Lazy quicksort

l45

Lazy quicksort
l Lazy evaluation can make some algorithms incremental, which 

can enormously improve their efficiency
l We show this with the quicksort algorithm

l Standard quicksort has an average time complexity of O(n log n) 
to sort n elements

l Lazy quicksort has a time complexity of O(n + k log k) to compute 
the k smallest elements out of n elements
l This is a very good bound!
l Furthermore, the value of k does not need to be known in advance.  

Elements can be computed incrementally until some condition is 
satisfied.

l To see how clever this is, try inventing the algorithm from scratch!

l46

Quicksort algorithm

l Pick a random 
element of L, the 
“pivot” X

l Partition into two 
sublists

l Recursively sort 
the sublists

l Append the results

L

L1 L2

S1 S2

S

< X ≥ X

recursive call recursive call

partition

append

l47

Quicksort example (on board)
l L = [7 3 2 8 6 4 1 9]
l Pivot = 7 (first element of L)
l L1 = [3 2 6 4 1], L2 = [7 8 9]
l …
l S1 = [1 2 3 4 6], S2 = [7 8 9]
l S = [1 2 3 4 6 7 8 9]

l48

Partition procedure
proc {Partition L X L1 L2}

case L of H|T then
if H<X then M1 in

L1=H|M1 {Partition T X M1 L2}
else /* H≥X */ M2 in

L2=H|M2 {Partition T X L1 M2}
end

[] nil then L1=nil L2=nil
end

end

l49

Append and quicksort
fun {Append L1 L2}

case L1 of H|T then H|{Append T L2}
[] nil then L2 end

end
fun {Quicksort L}

case L of X|M then L1 L2 S1 S2 in
{Partition L X L1 L2}
S1={Quicksort L1}
S2={Quicksort L2}
{Append S1 S2}

[] nil then nil
end

end

l50

Example eager execution
l Let us try to run this:

declare S in
S={Quicksort [4 3 2 5 6 4 3 2]}
{Browse S}

l What happens?
l Something is wrong!

l How do we fix this?
l A general rule when defining recursive functions!

l51

Append and quicksort (fixed)
fun {Append L1 L2}

case L1 of H|T then H|{Append T L2}
[] nil then L2 end

end
fun {Quicksort L}

case L of X|M then L1 L2 S1 S2 in
{Partition M X L1 L2}
S1={Quicksort L1} % L1 is strictly smaller than L
S2={Quicksort L2} % L2 is strictly smaller than L
{Append S1 X|S2}

[] nil then nil
end

end

M is strictly smaller than L

l52

Making quicksort lazy
l What has to be made lazy?

l Quicksort function becomes LQuicksort
l Append function becomes LAppend

l Partition is not lazy
l Sorting cannot work unless we look at all the 

elements of L
l Partition keeps the same eager definition
l We create the complete sublists L1 and L2

l53

Lazy append and quicksort
fun lazy {LAppend L1 L2}

case L1 of H|T then H|{LAppend T L2}
[] nil then L2 end

end
fun lazy {LQuicksort L}

case L of X|M then L1 L2 S1 S2 in
{Partition M X L1 L2}
S1={LQuicksort L1}
S2={LQuicksort L2}
{LAppend S1 X|S2}

[] nil then nil
end

end

l54

Example lazy executions
l Lazy append:

declare S in
S={LAppend [1 2 3] [4 5 6]}
{Browse S}
l What happens when asking for elements?

l Lazy quicksort:
declare S in
S={LQuicksort [4 3 2 5 6 4 3 2]}
{Browse S}
l What happens when asking for the first element?
l How much computation is done?  What is the time complexity?

l55

Execution steps…
l S={LQuicksort [2 3 4 1]} % Lazy suspension on S

{Browse S.1} 
% S is needed, so execute body of S={LQuicksort [2 3 4 1]}:
{Partition [3 4 1] 2 L1 L2}
S1={LQuicksort [1]} % Lazy suspension on S1
S2={LQuicksort [3 4]} % Lazy suspension on S2
S={LAppend S1 2|S2} % Lazy suspension on S

% S still needed, so execute body of LAppend:
case S1 of H|T then H|{LAppend T 2|S2}
[] nil then 2|S2 end

% S1 is needed, so execute body of S1={LQuicksort [1]}
{Partition nil 1 nil nil}
S1’={LQuicksort nil} % Lazy suspension on S1’
S2’={LQuicksort nil} % Lazy suspension on S2’
S1={LAppend S1’ 1|S2’} % Lazy suspension on S1

% S1 still needed, so execute body of LAppend:
case S1’ of H’|T’ then H’|{LAppend T’ 1|S2’}
[] nil then 1|S2’ end

% S1’ is needed, so execute body of S1’={LQuicksort nil}:
case nil of X’|M’ then (…)
[] nil then nil end
% Now we can do bindings:
S1’=nil

S1=1|S2’
S=1|{LAppend nil 2|S2}

{Browse (1|…).1}
% Displays 1

l Follow carefully what is happening
l When S is needed, it stays needed!
l We focus on the lazy suspensions

l S → {LQuicksort [2 3 4 1]}
S is needed, activates:
l S1 → {LQuicksort [1]}
l S2 → {LQuicksort [3 4]}
l S → {LAppend S1 2|S2}

S still needed, activates:
S1 is needed, activates:

§ S1’→{LQuicksort nil}
§ S2’→{LQuicksort nil}
§ S1→{LAppend S1’ 1|S2}

S1 still needed, activates:
S1’ is needed, activates:
§ S1’=nil

§ S1=1|S2’
l S=1|{LAppend nil 2|S2}

l

l56

Complexity of lazy quicksort

l To compute the smallest element, the number of operations is n + n/2 + n/4 + … + 1 
= 2n, so the time complexity is O(n)

l To compute the k smallest elements, a full “mini quicksort” is done as soon as the 
partitioned list has at least k elements, so the extra time complexity is O(k log k)

l Total time complexity is O(n + k log k)  

lPartition [3 4 1]

lPartition [1] lPartition [3 4]

Partition nil Partition nil

l1|…

nil 1|nil

l1|…

n traversal

n/2 traversal
(n/4, n/8, continues to 1)

l57

What is declarative 
programming?

l58

Declarative programming
l We have seen five functional paradigms

l Sequential functional programming
l Sequential functional programming with single assignment
l Deterministic dataflow
l Lazy evaluation
l Lazy deterministic dataflow

l We claim that they are all declarative
l What does this mean, exactly?
l Let us define it starting from the functional programming paradigm

l We show how to classify declarative paradigms according to their 
concepts and expressive power (Section 4.5.2 in the book)

l59

Functional programming
l All functional programs can be encoded as λ expressions
l Church-Rosser theorem:

l If ea reduces to eb (in 0 or more steps) and ea reduces to ec (in 0 or 
more steps), then there exists a term ed such that eb and ec can 
reduce to ed

l We say the λ calculus is confluent; it has the Church-Rosser property

ea → e2 → e3 → ⋯ → eb

ec → ec1 → ec2 → ⋯ → ed

↓
e2’
↓
e3’
↓
⋮

↓

⋮

↓
⋮

l60

Other functional paradigms?
l We see that functional programs are confluent

l The meaning is clear for the first paradigm, namely sequential 
functional programming

l But what does it mean for:
l Concurrency?
l Streams? (programs never terminate!)
l Single-assignment variables? (variables can be unbound!)

l We give a precise formal definition of “declarative 
programming” which covers these concepts
l Confluence: this handles concurrency (why?)
l Partial termination
l Equivalent stores

l61

I: Partial termination

l62

Partial termination

l Assume we have a concurrent 
agent with an input stream S1 
and an output stream S2

l It could execute as follows:
l S1=1|_ S2=2|_
l S1=1|2|_ S2=2|4|_
l S1=1|2|3|_ S2=2|4|6|_

l How is this functional?
l The program never terminates and 

the streams contain unbound 
variables

l With the right concepts, we 
can see this as functional 
execution:
l If S1 does not change, then 

S2 reaches a final value
l We call this “partial 

termination”
l We say the program has 

reached a “resting point”
l What about the unbound 

variables?

2 x
S1 S2

Agent

l63

II: Equivalent stores

l64

Single-assignment variables
l We claim that a functional program that uses single-

assignment variables is still functional
l Let’s see how to make this precise

l Consider the following program:
T1: thread X=foo(Z W) end
T2: thread Y=foo(Z W) end
T3: thread X=Y end
l Assume T1 and T2 execute before T3, then we have the store:

σ={x=foo(z w), y=foo(z w)}
l Assume T1 and T3 execute before T2, then we have the store:

σ’={x=foo(z w), y=x}
l How can we express that stores σ and σ’ are the same?

l65

A store is a logical formula
l Assume we have these two stores:

σ={x=foo(z w), y=foo(z w)}
σ’={x=foo(z w), y=x}

l The bindings of x and y are different for σ and σ’ but the possible 
values of x and y are the same in both stores
l Let’s see how to make this intuition precise

l A store σ corresponds to a relationship between values
l The store σ tells us that x is a record with label foo and arguments z and 

w, and that y is a record with label foo and arguments z and w
l For any values of x, y, z, and w, there are two possibilities: either they 

can be in the store σ or they cannot be in the store σ
l So the store σ is a logical formula, which can be true or false
l We write σ as a logical formula: σ ≡ x=foo(z w) ∧ y=foo(z w)

Interpretation and model
l Definition: An interpretation of a store σ

l An interpretation of a store σ is an assignment to all symbols in σ
l For all variables x in σ, assign a value x to x
l For all record symbols f in σ, assign a function f to f that has the same 

number of arguments as the record symbol and that returns a value
l Any interpretation of a store σ is either true or false

l A binding x=f(x1 … xn) is true if the value returned by f(x1 … xn) is equal 
to x; otherwise it is false

l A store σ = (x=f(x1 … xn) ∧… ∧ z=f(z1 … zn)) is true if all bindings are 
true, otherwise it is false

l Definition: A model of a store σ
l A model of σ is an interpretation in which σ is true

Equivalent stores
l Now we can define when two stores are equivalent

l Each store represents a logical formula that can be true or false
l Two stores are equivalent when, no matter how we assign values to their 

symbols, they are either both true or both false
l I.e., we cannot find values such that one store is true and the other is false

l We state this definition using the model concept
l We introduce the notation α⊨β which means “β is true in all models of α”

l Definition: Two stores σ and σ’ are logically equivalent if
l σ⊨σ’ and σ’⊨σ σ’ is true in all models of σ and σ is true in all models of σ’

l Another way to write this is:
l ⊨ (σ ⇔ σ’) (σ ⇔ σ’) is a tautology, i.e., it is true in all models

First-order logic
l To define equivalence of stores, we have introduced a 

little bit of first-order logic
l Logical formula (syntax)

l An expression that denotes a relationship between variables
l Model (semantics)

l A set of values and functions in which logical formulas are true 
or false

l If you want to understand more, you need to study first-
order logic!
l There are programming languages that are based on first-order 

logic, such as Prolog, constraint programming, and SQL

declarative 

programming

Definition of
declarative programming
l Now we can define precisely what declarative programming means

l A program is declarative if for all possible inputs:
l All executions for those inputs either:

l do not terminate, or
l all reach partial termination and give logically equivalent stores

l Remarks:
l “All executions” means all possible choices of the scheduler
l We say that a declarative program has “no observable nondeterminism”
l All five functional paradigms are declarative 

IV: Failure 

Fixing a buggy application
l Declarativeness is an extremely powerful property

l How do we write applications to be as declarative as possible?
l This is a major theme of the course! “All programs should be 

declarative except where they interact with the real world.”
l How do we fix an application that becomes nondeclarative?

l We can do failure confinement

l Nondeclarative behavior
l We will see later in the course that applications that interact with 

the real world can be nondeclarative
l That kind of nondeclarativeness is unavoidable but can be minimized

l Right now, let us see what happens when an application has a 
bug that makes it nondeclarative

Bugs are unavoidable
l “It is a truth universally acknowledged, that a program of a 

certain size must have bugs”
l With apologies to Jane Austen 😀

l Assume we have the following (simplified!) buggy program:
thread X=1 end
thread Y=2 end
thread X=Y end

l This program will always raise an exception
l Three stores are possible depending on the scheduler choices:

σ1={x=1,y=2}, σ2={x=1,y=1}, σ3={x=2,y=2}
l This is an observable nondeterminism, so it is nondeclarative

l We can fix this by doing failure confinement
l We will hide the nondeterminism from the rest of the program
l That way the program becomes declarative again

Failure confinement
l The program has three parts that can become 

inconsistent if there is a bug
l We use exceptions to protect these parts

thread try X1=1 S1=ok catch _ then S1=error end end
thread try Y1=2 S2=ok catch _ then S2=error end end
thread try X1=Y1 S3=ok catch _ then S3=error end end
if S1==error orelse S2==error orelse S3==error then

X=1  Y=1  /* default result when there is an error */
else

X=X1  Y=Y1  /* correct result when there is no error */
end

Table of
declarative paradigms

Declarative paradigms

Conclusions

Conclusions
l Lazy evaluation

l Functions are evaluated only if their results are needed
l This extends deterministic dataflow with the WaitNeeded operation
l Programs can use infinite lists and be made more incremental
l Lazy evaluation can be combined with concurrency

l Declarative programming
l An application should be declarative except for real-world interaction
l We need to define precisely what is declarative programming

l We give a precise definition of declarative programming using the concepts of 
confluence, partial termination, and logical equivalence

l Declarativeness is an observational concept: a program can behave 
declaratively even if it is written in a nondeclarative paradigm

l Next lecture: Advanced declarative algorithm design
l Declarative algorithms can be as efficient as nondeclarative algorithms

